
1 Overview

This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2019-06_D20wk38 related to the title of this CR.

Contact:
Kalahasti, Narendra

Narendra.Kalahasti@anritsu.com
2 Corrections required
2.1 Change 1

	Function name
	fl_TC_8_1_4_2_1_1_NR5GC_EUTRA_TestBody

	Reason for change
	1. In the current TTCN implementation, assigning values to variables v_DRBIdList, v_DRBId, and v_RbId_DRB at the start of the test body will cause run time error. DRBInfoList at EUTRA PTC will be populated after the f_Encode_RRCConnectionReconfiguration_HO5GToEUTRA function.
2. In case of Single PDN (Internet only) operation, NR PTC will use PDN_3 as default during NR preamble. However, at EUTRA_PTC, f_EUTRA_CheckDataPath is currently hardcoded to use PDN_2. This causes a mismatch for the PDN IP addresses used during checking of data path.
3. f_EUTRA_CheckDataPath should be executed for both pc_IP_Ping = TRUE or FALSE.

	Summary of change
	1. Asign values to v_DRBIdList, v_DRBId, and v_RbId_DRB before checking of data path at Step 6.
2. Get PDU Session Info from NR PTC to get PDN index information.
3. Remove if condition so that checking of data path will be executed regardless of pc_IP_Ping = TRUE or FALSE.

	TTCN module
	IRAT_Handover_NR5GC_EUTRA

	MCC160 Comment
	


Before change:
  function fl_TC_8_1_4_2_1_1_NR5GC_EUTRA_TestBody() runs on EUTRA_Multi5GS_PTC

  {

    var EUTRA_NR_CoOrd_GUTI_Type v_CoOrd_Guti;

    var EUTRA_NR_CoOrd_Security_Type v_SecurityRx;

    var template (value) MobileIdentity v_MappedGUTI;

    var SRB_COMMON_IND v_ReceivedMsg;

    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_Cell1);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId(v_GutiParams.PLMN_Identity);

    var TrackingAreaCode v_Tac := f_EUTRA_CellInfo_GetTAC(eutra_Cell1);

    var template (present) TrackingAreaId v_TAI := cr_TAI(v_PLMN, bit2oct(v_Tac));

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType (NORMAL);

    var EUTRA_SecurityParams_Type v_Auth_Params := f_EUTRA_Security_Get();

    var template (value) EUTRA38_DRBInfoList_Type v_DRBIdList := f_EUTRA38_MobileInfo_GetDRBInfoList();

    var integer v_DRBId := valueof(v_DRBIdList[0].DRB_Id);

    var template (value) RadioBearerId_Type v_RbId_DRB := cs_RadioBearerId_DRB(v_DRBId);
    var octetstring v_HOCmd := f_Encode_RRCConnectionReconfiguration_HO5GToEUTRA(eutra_Cell1,

                                                                                 f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell1),

                                                                                 f_EUTRA_CellInfo_GetDL_ChBandwidth (eutra_Cell1),

                                                                                 f_EUTRA_CellInfo_GetUL_ChBandwidth (eutra_Cell1),

                                                                                 f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1),

                                                                                 f_EUTRA_CellInfo_GetRootSequenceIndex (eutra_Cell1));

    v_CoOrd_Guti := f_EUTRA_NR_WaitForCoOrd_GutiParams(NR);

    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger); // @sic R5s200509 sic@

    v_SecurityRx := f_EUTRA_NR_WaitForCoOrd_Security (NR);

    v_MappedGUTI := f_EUTRA38_NR_Convert5G_CoOrdToEPS_GUTI (v_CoOrd_Guti, omit);

    // Configure C-RNTI based contention resolution in target cell

    f_EUTRA_SS_ConfigRachProcedure(eutra_Cell1, tsc_C_RNTI_Def2, cs_RachProcedureConfig_CRNTI_Dedicated(f_EUTRA_CellInfo_GetRAR_TA(eutra_Cell1),

                                                                                                        f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1))); // @sic R5-123712 sic@

    f_EUTRA_CellInfo_SetC_RNTI ( eutra_Cell1, tsc_C_RNTI_Def2 );

    // activate RRC security at SS target cell for SRB 1, 2 and 1 AM DRB

    v_Auth_Params := f_EUTRA38_MappedContextFromN1_InitNAS(eutra_Cell1, v_SecurityRx, true);

    v_Auth_Params.AS_Ciphering.ActTimeList := f_EUTRA_RRC_CipherActTime_GetForHO(eutra_Cell1, tsc_DrbList_Def);

    // @sic R5s200938 sic@

    f_EUTRA_SS_RRC_EnableIntProt_CiphULandDL(eutra_Cell1,

                                             v_Auth_Params.AS_Integrity,

                                             v_Auth_Params.AS_Ciphering,

                                             cs_TimingInfo_Now);

    f_EUTRA_Security_Set(v_Auth_Params);

    //Configure UL grant configuration ("OnSR", periodic TA is NOT started) in target cell (cell 1)

    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_ULGrantAllocation_Def_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_UplinkTimeAlignment_Synch_None));

    // Send EUTRA HO command to NR

    f_EUTRA_NR_SendCoOrd(NR, cms_EUTRA_NR_OctetData (v_HOCmd, "HO Cmd"));

    // @siclog "Step 2" siclog@

    SRB.receive( car_SRB1_RrcPdu_IND( eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete( tsc_RRC_TI_Def ) ) );

    /* SS automatically checks that UE transmit an RRCConnectionReconfigurationComplete using the indicated

       security key and the indicated security algorithms */

    f_EUTRA_PreliminaryPass( __FILE__, __LINE__, "Step 2" );

    //Configure UL grant configuration

    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_PUCCH_Synch_Config_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_UplinkTimeAlignment_Synch_Auto_Def)); // @sic R5s191111 sic@

    // @siclog "Step 3" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                      cr_TAU_Request_N1ToS1((tsc_EpsUpdate_TaUpdate, tsc_EpsUpdate_Combined_IMSIAttach, tsc_EpsUpdate_Combined_TaLaUpdate),  // @sic R5-201638 sic@

                                                                            v_SecurityRx.KSI,  // @sic R5s200509 sic@

                                                                            cr_DRXparameter_Any ('5C'O) ifpresent,

                                                                            v_AdditionalUpdateType,

                                                                            v_MappedGUTI,

                                                                            v_TAI ifpresent,  // @sic R5s200509 sic@

                                                                            -,

                                                                            tsc_NasKsi_MappedSecurityContext,  // @sic R5-201215, R5s200509 sic@

                                                                            cdr_NAS_KeySetIdentifier_tlv (?, ?) ifpresent,

                                                                            -, -,

                                                                            cr_508_EPS_BearerContextStatusTAUReq ifpresent))))-> value v_ReceivedMsg; // @sic R5s200509, R5s200968, Rs5200760 sic@

    // @siclog "Step 4-5" siclog@

    f_EUTRA38_TrackingAreaUpdateFromN1_WithoutReq (eutra_Cell1,

                                                   v_ReceivedMsg.Signalling.Nas[0],

                                                   -, noRrcConnectionRelease,

                                                   -, true);   // @sic R5s191111 change 9 sic@

    // @siclog "Step 6" siclog@

    if (pc_IP_Ping) { // @sic R5-204452 sic@

      f_EUTRA_CheckDataPath(eutra_Cell1, PDCP, defaultBearer, "Step 6", v_RbId_DRB);
    }
  }
After change:
  import from Common4G5G all;
  …

  function fl_TC_8_1_4_2_1_1_NR5GC_EUTRA_TestBody() runs on EUTRA_Multi5GS_PTC

  {

    var EUTRA_NR_CoOrd_GUTI_Type v_CoOrd_Guti;

    var EUTRA_NR_CoOrd_Security_Type v_SecurityRx;

    var template (value) MobileIdentity v_MappedGUTI;

    var SRB_COMMON_IND v_ReceivedMsg;

    var GutiParameters_Type v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_Cell1);

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId(v_GutiParams.PLMN_Identity);

    var TrackingAreaCode v_Tac := f_EUTRA_CellInfo_GetTAC(eutra_Cell1);

    var template (present) TrackingAreaId v_TAI := cr_TAI(v_PLMN, bit2oct(v_Tac));

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType (NORMAL);

    var EUTRA_SecurityParams_Type v_Auth_Params := f_EUTRA_Security_Get();

    var template (value) EUTRA38_DRBInfoList_Type v_DRBIdList;

    var integer v_DRBId;

    var template (value) RadioBearerId_Type v_RbId_DRB;
    var EUTRA_NR_PduSessionInfoList_Type v_PduSessionInfoList;
    var octetstring v_HOCmd := f_Encode_RRCConnectionReconfiguration_HO5GToEUTRA(eutra_Cell1,

                                                                                 f_EUTRA_CellInfo_GetFrequencyIEs(eutra_Cell1),

                                                                                 f_EUTRA_CellInfo_GetDL_ChBandwidth (eutra_Cell1),

                                                                                 f_EUTRA_CellInfo_GetUL_ChBandwidth (eutra_Cell1),

                                                                                 f_EUTRA_CellInfo_GetPhyCellId(eutra_Cell1),

                                                                                 f_EUTRA_CellInfo_GetRootSequenceIndex (eutra_Cell1));

    v_CoOrd_Guti := f_EUTRA_NR_WaitForCoOrd_GutiParams(NR);

    f_EUTRA_NR_SendCoOrd (NR, cms_EUTRA_NR_Trigger); // @sic R5s200509 sic@

    v_SecurityRx := f_EUTRA_NR_WaitForCoOrd_Security (NR);

    v_MappedGUTI := f_EUTRA38_NR_Convert5G_CoOrdToEPS_GUTI (v_CoOrd_Guti, omit);

    // Configure C-RNTI based contention resolution in target cell

    f_EUTRA_SS_ConfigRachProcedure(eutra_Cell1, tsc_C_RNTI_Def2, cs_RachProcedureConfig_CRNTI_Dedicated(f_EUTRA_CellInfo_GetRAR_TA(eutra_Cell1),

                                                                                                        f_EUTRA_CellInfo_GetDL_ChBandwidth(eutra_Cell1))); // @sic R5-123712 sic@

    f_EUTRA_CellInfo_SetC_RNTI ( eutra_Cell1, tsc_C_RNTI_Def2 );

    // activate RRC security at SS target cell for SRB 1, 2 and 1 AM DRB

    v_Auth_Params := f_EUTRA38_MappedContextFromN1_InitNAS(eutra_Cell1, v_SecurityRx, true);

    v_Auth_Params.AS_Ciphering.ActTimeList := f_EUTRA_RRC_CipherActTime_GetForHO(eutra_Cell1, tsc_DrbList_Def);

    // @sic R5s200938 sic@

    f_EUTRA_SS_RRC_EnableIntProt_CiphULandDL(eutra_Cell1,

                                             v_Auth_Params.AS_Integrity,

                                             v_Auth_Params.AS_Ciphering,

                                             cs_TimingInfo_Now);

    f_EUTRA_Security_Set(v_Auth_Params);

    //Configure UL grant configuration ("OnSR", periodic TA is NOT started) in target cell (cell 1)

    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_ULGrantAllocation_Def_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_UplinkTimeAlignment_Synch_None));

    // Send EUTRA HO command to NR

    f_EUTRA_NR_SendCoOrd(NR, cms_EUTRA_NR_OctetData (v_HOCmd, "HO Cmd"));

    // @siclog "Step 2" siclog@

    SRB.receive( car_SRB1_RrcPdu_IND( eutra_Cell1, cr_508_RRCConnectionReconfigurationComplete( tsc_RRC_TI_Def ) ) );

    /* SS automatically checks that UE transmit an RRCConnectionReconfigurationComplete using the indicated

       security key and the indicated security algorithms */

    f_EUTRA_PreliminaryPass( __FILE__, __LINE__, "Step 2" );

    //Configure UL grant configuration

    f_EUTRA_SS_CommonCellConfig(eutra_Cell1, cas_PUCCH_Synch_Config_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_UplinkTimeAlignment_Synch_Auto_Def)); // @sic R5s191111 sic@

    // @siclog "Step 3" siclog@

    SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                      cr_TAU_Request_N1ToS1((tsc_EpsUpdate_TaUpdate, tsc_EpsUpdate_Combined_IMSIAttach, tsc_EpsUpdate_Combined_TaLaUpdate),  // @sic R5-201638 sic@

                                                                            v_SecurityRx.KSI,  // @sic R5s200509 sic@

                                                                            cr_DRXparameter_Any ('5C'O) ifpresent,

                                                                            v_AdditionalUpdateType,

                                                                            v_MappedGUTI,

                                                                            v_TAI ifpresent,  // @sic R5s200509 sic@

                                                                            -,

                                                                            tsc_NasKsi_MappedSecurityContext,  // @sic R5-201215, R5s200509 sic@

                                                                            cdr_NAS_KeySetIdentifier_tlv (?, ?) ifpresent,

                                                                            -, -,

                                                                            cr_508_EPS_BearerContextStatusTAUReq ifpresent))))-> value v_ReceivedMsg; // @sic R5s200509, R5s200968, Rs5200760 sic@

    // @siclog "Step 4-5" siclog@

    f_EUTRA38_TrackingAreaUpdateFromN1_WithoutReq (eutra_Cell1,

                                                   v_ReceivedMsg.Signalling.Nas[0],

                                                   -, noRrcConnectionRelease,

                                                   -, true);   // @sic R5s191111 change 9 sic@

    v_DRBIdList := f_EUTRA38_MobileInfo_GetDRBInfoList();

    v_DRBId := valueof(v_DRBIdList[0].DRB_Id);

    v_RbId_DRB := cs_RadioBearerId_DRB(v_DRBId);
    v_PduSessionInfoList := f_EUTRA_NR_WaitForCoOrd_PduSessionInfoList(NR);
    // @siclog "Step 6" siclog@

    //REMOVED if (pc_IP_Ping) { // @sic R5-204452 sic@

    f_EUTRA_CheckDataPath(eutra_Cell1, PDCP, defaultBearer, "Step 6", v_RbId_DRB, -, -, valueof(v_PduSessionInfoList[0].PdnIndex));
    // REMOVED }
  }
2.2 Change 2

	Function name
	fl_TC_8_1_4_2_1_1_TestBody

	Reason for change
	As stated in Change 1, PDU Session Info will be passed to EUTRA_PTC before checking data path at Step 6.

	Summary of change
	Pass PDU Session Info to EUTRA_PTC.

	TTCN module
	IRAT_Handover_NR5GC_NR

	MCC160 Comment
	


Before change:
  function fl_TC_8_1_4_2_1_1_TestBody() runs on NR5GC_PTC

  {

    var octetstring v_EUTRA_HOCmd;

    var NasCountInfo_Type v_NASCountInfo;

   // @sic R5-2001215, R5-202640 sic@

    //@siclog "Step 1" siclog@

    // Send both authentication and NR sysinfo paramters to EUTRA ready for HO

    v_NASCountInfo := f_NR5GC_TrackingAreaUpdateFromN1(nr_Cell1);

    v_EUTRA_HOCmd := f_EUTRA_NR_WaitForCoOrd_Data(EUTRA, "HO Cmd"); // @sic R5s200509 sic@

    SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, -, cs_MobilityFromNRCommand (tsc_NR_RRC_TI_Def, v_EUTRA_HOCmd, substr(v_NASCountInfo.DL, 3, 1) )));

    //@siclog "Step 2" siclog@

    f_EUTRA_NR_WaitForCoOrd_Trigger (EUTRA, "Postamble");

  }
After change:
  function fl_TC_8_1_4_2_1_1_TestBody() runs on NR5GC_PTC

  {

    var octetstring v_EUTRA_HOCmd;

    var NasCountInfo_Type v_NASCountInfo;

   // @sic R5-2001215, R5-202640 sic@

    //@siclog "Step 1" siclog@

    // Send both authentication and NR sysinfo paramters to EUTRA ready for HO

    v_NASCountInfo := f_NR5GC_TrackingAreaUpdateFromN1(nr_Cell1);

    v_EUTRA_HOCmd := f_EUTRA_NR_WaitForCoOrd_Data(EUTRA, "HO Cmd"); // @sic R5s200509 sic@

    SRB.send(cas_NR_SRB1_RrcPdu_REQ(nr_Cell1, -, cs_MobilityFromNRCommand (tsc_NR_RRC_TI_Def, v_EUTRA_HOCmd, substr(v_NASCountInfo.DL, 3, 1) )));

    //@siclog "Step 2" siclog@

    f_NR5GC_SendPduSessionInfoListToEUTRA();
    f_EUTRA_NR_WaitForCoOrd_Trigger (EUTRA, "Postamble");

  }
2.3 Change 3
	Function name
	f_EUTRA_CheckDataPath

	Reason for change
	As stated in Change 1, PDN index should not be hardcoded to PDN_2 inside f_EUTRA_CheckDataPath function.

	Summary of change
	Add 8th parameter to f_EUTRA_CheckDataPath for the PDN index type.

	TTCN module
	EUTRA38_CommonProcedures

	MCC160 Comment
	


Before change:
  function f_EUTRA_CheckDataPath(EUTRA_CellId_Type p_CellId,

                                 PDCP_Protocol_Type p_PDCP_Protocol,

                                 DataPathCheck_BearerType_Type p_BearerType,

                                 charstring p_StepNo,

                                 template (value) RadioBearerId_Type p_DRB,

                                 integer p_SequenceNumber := 1,

                                 boolean p_UsePING := true) runs on EUTRA_PTC

  {

   var PDN_Index_Type v_PdnIndex := PDN_2;

   var octetstring v_PdcpSduTX := f_LoopbackModeBorPing_GetDataTX(v_PdnIndex, p_BearerType, p_SequenceNumber, p_UsePING);

   var PDCP_SDUList_Type v_PDCP_SDUList := { v_PdcpSduTX };

   var template (value) L2DataList_Type v_PDCP_DataTX;

   var template (present) L2DataList_Type v_PDCP_DataRX;

   var DRB_COMMON_IND v_DRB_COMMON_IND;

   var octetstring v_PdcpSduRX;

   select (p_PDCP_Protocol) {

     case (PDCP) {

       v_PDCP_DataTX := cs_L2DataList_PdcpSdu(v_PDCP_SDUList);

       v_PDCP_DataRX := cr_L2DataList_PdcpSdu;

     }

     case (NR_PDCP) {

       v_PDCP_DataTX := cs_L2DataList_NrPdcpSdu(v_PDCP_SDUList);

       v_PDCP_DataRX := cr_L2DataList_NrPdcpSdu;

     }

   }

   if (pc_IP_Ping) { f_IP_Routing_ChangeToRawmode(IP); }   /* @sic R5s200044 change 4 sic@ */

   DRB.send(cas_DRB_COMMON_REQ(p_CellId, p_DRB, cs_TimingInfo_Now, v_PDCP_DataTX)); // @sic R5s200514 sic@

   DRB.receive(car_DRB_COMMON_IND(p_CellId, p_DRB, cr_TimingInfo_Any, v_PDCP_DataRX)) -> value v_DRB_COMMON_IND; // @sic R5s200514 sic@

   if (pc_IP_Ping) { f_IP_Routing_RestoreRoutingTable(IP); }   /* @sic R5s200044 change 4 sic@ */

   select (p_PDCP_Protocol) {

     case (PDCP)    { v_PdcpSduRX := v_DRB_COMMON_IND.U_Plane.SubframeData.PduSduList.PdcpSdu[0]; }

     case (NR_PDCP) { v_PdcpSduRX := v_DRB_COMMON_IND.U_Plane.SubframeData.PduSduList.NrPdcpSdu[0]; }

   }

   if (not f_LoopbackModeBorPing_CheckDataRX(v_PdcpSduTX, v_PdcpSduRX)) {

     f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid Echo Reply");

   } else {

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, p_StepNo);

   }

  }
After change:
  function f_EUTRA_CheckDataPath(EUTRA_CellId_Type p_CellId,

                                 PDCP_Protocol_Type p_PDCP_Protocol,

                                 DataPathCheck_BearerType_Type p_BearerType,

                                 charstring p_StepNo,

                                 template (value) RadioBearerId_Type p_DRB,

                                 integer p_SequenceNumber := 1,

                                 boolean p_UsePING := true,

                                 PDN_Index_Type p_PdnIndex := PDN_2) runs on EUTRA_PTC

  {

   var PDN_Index_Type v_PdnIndex := p_PdnIndex;

   var octetstring v_PdcpSduTX := f_LoopbackModeBorPing_GetDataTX(v_PdnIndex, p_BearerType, p_SequenceNumber, p_UsePING);

   var PDCP_SDUList_Type v_PDCP_SDUList := { v_PdcpSduTX };

   var template (value) L2DataList_Type v_PDCP_DataTX;

   var template (present) L2DataList_Type v_PDCP_DataRX;

   var DRB_COMMON_IND v_DRB_COMMON_IND;

   var octetstring v_PdcpSduRX;

   select (p_PDCP_Protocol) {

     case (PDCP) {

       v_PDCP_DataTX := cs_L2DataList_PdcpSdu(v_PDCP_SDUList);

       v_PDCP_DataRX := cr_L2DataList_PdcpSdu;

     }

     case (NR_PDCP) {

       v_PDCP_DataTX := cs_L2DataList_NrPdcpSdu(v_PDCP_SDUList);

       v_PDCP_DataRX := cr_L2DataList_NrPdcpSdu;

     }

   }

   if (pc_IP_Ping) { f_IP_Routing_ChangeToRawmode(IP); }   /* @sic R5s200044 change 4 sic@ */

   DRB.send(cas_DRB_COMMON_REQ(p_CellId, p_DRB, cs_TimingInfo_Now, v_PDCP_DataTX)); // @sic R5s200514 sic@

   DRB.receive(car_DRB_COMMON_IND(p_CellId, p_DRB, cr_TimingInfo_Any, v_PDCP_DataRX)) -> value v_DRB_COMMON_IND; // @sic R5s200514 sic@

   if (pc_IP_Ping) { f_IP_Routing_RestoreRoutingTable(IP); }   /* @sic R5s200044 change 4 sic@ */

   select (p_PDCP_Protocol) {

     case (PDCP)    { v_PdcpSduRX := v_DRB_COMMON_IND.U_Plane.SubframeData.PduSduList.PdcpSdu[0]; }

     case (NR_PDCP) { v_PdcpSduRX := v_DRB_COMMON_IND.U_Plane.SubframeData.PduSduList.NrPdcpSdu[0]; }

   }

   if (not f_LoopbackModeBorPing_CheckDataRX(v_PdcpSduTX, v_PdcpSduRX)) {

     f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Invalid Echo Reply");

   } else {

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, p_StepNo);

   }

  }
